

Generic Programming with Multiple Parameters

José Pedro Magalhães

Department of Computer Science, University of Oxford, Oxford, UK
jpm@cs.ox.ac.uk

Abstract. Generic programming, a form of abstraction in programming languages that serves to reduce code duplication by exploiting the regular structure of algebraic datatypes, has been present in the Haskell language in different forms for many years. Lately, a library for generic deriving has been given native support in the compiler, allowing programmers to write functions such as *fmap* that abstract over one datatype parameter generically. The power of this approach is limited to dealing with one parameter per datatype, however. In this paper, we lift this restriction by providing a generalisation of generic deriving that supports multiple parameters, making essential use of datatype promotion and kind polymorphism. We show example encodings of datatypes, and how to define a map function that operates on multiple parameters simultaneously.

1 Introduction

Haskell [12], a pure, lazy, strongly-typed functional programming language, has been a research vehicle for generic programming almost since its inception [2]. Generic programming is a form of abstraction which exploits the structure of algebraic datatypes in order to define functionality that operates on many datatypes uniformly, reducing code duplication. With generic programming, certain functions (such as data serialisation and traversals) can be written once and for all, working for existing and future types of data.

Early approaches to generic programming in Haskell were separate extensions to the compiler or preprocessors [3], but recently all approaches are bundled as libraries [13], sometimes with direct compiler support [11]. The easier availability of generic programming functionality appears to increase its usage; in particular, offering native support for an approach in the compiler confers a sense of stability to a specific approach, allowing more programmers to use and benefit from generic programming.

This switch to library approaches simplifies the task of the generic programming library developer, since it is typically easier to maintain a library than a separate extension. It also presents the library user with fewer barriers to adoption, as it requires no external tools. However, it can compromise expressivity and usability, as library approaches are limited to the capabilities of the Haskell language itself. Performance and quality of error messages are common complaints of generic programming libraries, but also reduced support for certain datatypes can be a concern. In particular, the Generic Haskell compiler [6], an early, pre-processor approach, had full support for generic functions abstracting over multiple parameters. The same goes for the generics in the Clean language [1], which were implemented in the kind-indexed style of

Generic Haskell. However, to our knowledge no library approach to generic programming in Haskell has native support for abstraction over multiple datatype parameters. Concretely, consider the following example:

```

data WTree  $\alpha$   $\omega$  = Leaf  $\alpha$ 
                | Fork (WTree  $\alpha$   $\omega$ ) (WTree  $\alpha$   $\omega$ )
                | WithWeight (WTree  $\alpha$   $\omega$ )  $\omega$ 

mapWTree :: ( $\alpha \rightarrow \alpha'$ )  $\rightarrow$  ( $\omega \rightarrow \omega'$ )  $\rightarrow$  WTree  $\alpha$   $\omega$   $\rightarrow$  WTree  $\alpha'$   $\omega'$ 
mapWTree f g (Leaf a)           = Leaf (f a)
mapWTree f g (Fork l r)         = Fork (mapWTree f g l) (mapWTree f g r)
mapWTree f g (WithWeight t w) = WithWeight (mapWTree f g t) (g w)

```

Clearly, functions like *mapWTree* follow the structure of the datatype, and should be defined generically, once and for all. In this paper we focus on the problem of defining *functions* such as *mapWTree* generically; many (if not all) modern generic programming approaches support the *WTree* datatype, but not generic operations such as *mapWTree*. The only exception that we are aware of is a convoluted implementation using Scrap Your Boilerplate [5], which relies on runtime type comparison, casting, and (virtual) serialisation.¹ Our approach focuses instead on encoding the parameters adequately in the generic representation.

In this paper, we focus our attention on the generic-deriving approach, as implemented in the Glasgow Haskell Compiler (GHC) [8, Chapter 11]. To balance complexity and ease-of-use, generic-deriving was designed from the start with one compromise in mind: many generic functions (e.g. *fmap* and *traverse*) abstract over one datatype parameter (i.e. they operate on *type containers*, or types of kind $\star \rightarrow \star$), but few require abstraction over more than one parameter. As such, the design of generic-deriving could be kept relatively simple, at the loss of some generality. Recently, with the advent of kind polymorphism in GHC [16], many libraries are being generalised from kind-specific to kind-polymorphic (e.g. *Typeable* [16]). In this paper we describe an elegant generalisation of generic-deriving that works with multiple parameters, lifting the one-parameter restriction without requiring the full-blown power (and complexity) of indexed functors [7]. Our solution is based on an earlier, failed attempt [9, Section 6.1], but uses new insights to overcome old challenges: we use heterogeneous collections [4] to encode a function with multiple arguments as a regular single-argument function that takes a list of arguments.

In the remainder of this paper, we first review one-parameter generic-deriving (Section 2). We proceed to describe our generalisation in Section 3, showing example datatype encodings (Section 3.2), and a generalised map function (Section 3.3). We list limitations and propose future work in Section 4, and conclude in Section 5.

Notation In order to avoid syntactic clutter and to help the reader, we adopt a liberal Haskell notation in this paper. Datatype promotion makes datatype definitions *also* define a kind (with the name of the datatype being defined), while the constructors *also* become the types that inhabit that kind; we assume the existence of a **kind** keyword, which allows us to define kinds that behave as if they had arisen from datatype promotion, ex-

¹ <http://okmij.org/ftp/Haskell/generics.html#gmap>

cept that they do not define a datatype and constructors. This helps us preventing name clashes. We omit the keywords **type family** and **type instance** entirely, making type-level functions look like their value-level counterparts. We colour constructors in *blue*, types in *red*, and kinds in *green*. Promoted lists are prefixed with a quote, to distinguish them from the list type. Additionally, we use Greek letters for type variables, apart from κ , which is reserved for kind variables. A grayscale version of this paper is available at http://dreixel.net/research/pdf/gmpm_nocolour.pdf. The syntactic sugar is only for presentation purposes. An executable version of the code, which compiles with GHC 7.6.2, is available at <http://dreixel.net/research/code/gmpm.zip>.

2 Generic programming with one parameter

In this section we review the generic-deriving library (in the slightly revised version of Magalhães and Löh [10]), which supports abstraction over (at most) one datatype parameter. We omit meta-information, as it is not relevant to our development.

2.1 Universe

Using datatype promotion and kind polymorphism, we can keep the “realm” of generic representations of user datatypes separate from the realm of user datatypes by defining a new kind. Like types classify values, kinds classify types. User datatypes always have kind \star . For the generic representation, we define a kind *Univ* that aggregates the types used to represent user datatypes in generic-deriving. *U* encodes constructors without arguments, *P* an occurrence of the parameter, *K* a type that does not contain the parameter, *R* a type that contains the parameter, $(:+:)$ the choice between two constructors, $(:\times:)$ is used for constructors with multiple arguments, and $(:@:)$ encodes the application of a functor to a type. The interpretation datatype *In v ρ* encodes the values associated with a representation type *v* and parameter ρ :

kind <i>Univ</i> =	data <i>In</i> (<i>v</i> :: <i>Univ</i>) (ρ :: \star) :: \star where
<i>U</i>	<i>U_I</i> :: <i>In U</i> ρ
<i>P</i>	<i>Par_I</i> :: ρ → <i>In P</i> ρ
<i>K</i> \star	<i>K_I</i> :: α → <i>In (K α)</i> ρ
<i>R</i> ($\star \rightarrow \star$)	<i>Rec_I</i> :: $\phi \rho$ → <i>In (R φ)</i> ρ
<i>Univ</i> $+:$ <i>Univ</i>	<i>L_I</i> :: <i>In φ ρ</i> → <i>In (φ $+:$ ψ)</i> ρ
<i>Univ</i> \times <i>Univ</i>	<i>R_I</i> :: <i>In ψ ρ</i> → <i>In (φ $+:$ ψ)</i> ρ
$\star \rightarrow \star$ $:@$ <i>Univ</i>	$(:\times:)$:: <i>In φ ρ</i> → <i>In ψ ρ</i> → <i>In (φ \times ψ)</i> ρ
	<i>App_I</i> :: ϕ (<i>In ψ ρ</i>) → <i>In (φ $:@$ ψ)</i> ρ

There is some redundancy in the universe codes; for example, *R φ* is essentially a shortcut for $\phi :@: P$, but the latter has a more complicated representation (ϕ (*In P* ρ)) instead of just $\phi \rho$. Our new encoding in Section 3 also solves this issue.

The *Generic* type class mediates between user datatypes and their generic representation. A *Rep* type function is used to encode the generic representation of some user datatype α , while *Par* identifies the parameter. Conversion functions *from* and *to* witness the isomorphism between α and *In (Rep α) (Par α)*:

```

class Generic ( $\alpha :: \star$ ) where
  Rep  $\alpha :: Univ$ 
  Par  $\alpha :: \star$ 
  from ::  $\alpha \rightarrow In (Rep\ \alpha) (Par\ \alpha)$ 
  to   ::  $In (Rep\ \alpha) (Par\ \alpha) \rightarrow \alpha$ 

```

Instances of this class are trivial, but tedious to write, and are necessary for each datatype intended to be used generically. Fortunately, these instances are automatically derivable by GHC, therefore making generic-deriving a “built-in” generic programming approach.

2.2 Datatype encodings

The universe of generic-deriving is best understood through sample datatype instantiations, which we provide in this section. We start with the encoding of lists. A list is a choice ($(:+:)$) between an empty constructor (U) or two ($(:\times:)$) arguments—the parameter (P) and another list ($R []$):

```

instance Generic [ $\alpha$ ] where
  Rep [ $\alpha$ ] =  $U :+: P :\times: R []$ 
  Par [ $\alpha$ ] =  $\alpha$ 
  from []   =  $L_1 U_1$ 
  from ( $h : t$ ) =  $R_1 (Par_1 h :\times: Rec_1 t)$ 
  to ( $L_1 U_1$ )           = []
  to ( $R_1 (Par_1 h :\times: Rec_1 t)$ ) =  $h : t$ 

```

From this point forward we shall omit the *to* function, as it is always entirely symmetric to *from*.

A slightly more complicated encoding is that of rose (or multiway) trees. Since we use application of lists in the representation, the conversion functions need to be able to map over these lists. This is achieved using the *fmap* function, a generic map that can be defined for all *Generic* types (Section 2.3):

```

data RTree  $\alpha = RTree\ \alpha [RTree\ \alpha]$ 
instance Generic (RTree  $\alpha$ ) where
  Rep (RTree  $\alpha$ ) =  $P :\times: ([] :@: R\ RTree)$ 
  Par (RTree  $\alpha$ ) =  $\alpha$ 
  from (RTree  $x\ xs$ ) =  $Par_1\ x :\times: App_1 (fmap\ Rec_1\ xs)$ 

```

2.3 Mapping

As generic-deriving supports abstraction over one parameter, we can define the standard function *fmap* generically. The user-facing class for this function abstracts over type containers ϕ of kind $\star \rightarrow \star$:

```

class Functor ( $\phi :: \star \rightarrow \star$ ) where
  fmap ::  $(\alpha \rightarrow \beta) \rightarrow \phi\ \alpha \rightarrow \phi\ \beta$ 

```

The generic version, $fmap_R$, on the other hand, operates on the generic representation. We need to give an instance of $fmap_R$ for each representation type; we use another type class for this purpose:

```
class FunctorR (v :: Univ) where
  fmapR :: ( $\alpha \rightarrow \beta$ )  $\rightarrow$  In v  $\alpha \rightarrow$  In v  $\beta$ 
```

Sums, products, units, and constants are trivial:

```
instance FunctorR U where
  fmapR - UI = UI
instance FunctorR (K  $\alpha$ ) where
  fmapR - (KI x) = KI x
instance (FunctorR  $\phi$ , FunctorR  $\psi$ )  $\Rightarrow$  FunctorR ( $\phi$  :+ :  $\psi$ ) where
  fmapR f (LI x) = LI (fmapR f x)
  fmapR f (RI x) = RI (fmapR f x)
instance (FunctorR  $\phi$ , FunctorR  $\psi$ )  $\Rightarrow$  FunctorR ( $\phi$  : $\times$  :  $\psi$ ) where
  fmapR f (x : $\times$  : y) = fmapR f x : $\times$  : fmapR f y
```

More interesting are the instances for: the parameter, where we apply the mapping function; recursion into type containers, where we recurse using $fmap$; and application of type containers, where we $fmap$ on the outer container, and $fmap_R$ on the inner representation:

```
instance FunctorR P where
  fmapR f (ParI x) = ParI (f x)
instance (Functor  $\phi$ )  $\Rightarrow$  FunctorR (R  $\phi$ ) where
  fmapR f (RecI x) = RecI (fmap f x)
instance (Functor  $\phi$ , FunctorR v)  $\Rightarrow$  FunctorR ( $\phi$  :@ : v) where
  fmapR f (AppI x) = AppI (fmap (fmapR f) x)
```

Providing instances for types with a *Generic* instance, such as $[\alpha]$, requires only using $fmap_R$ and converting *from/to* the original datatype:

```
instance Functor [] where
  fmap f = to  $\circ$  fmapR f  $\circ$  from
```

We now have an easy way to define a generic $fmap$ for any user datatype with a *Generic* instance.

3 Generic programming with multiple parameters

Having seen the current implementation of generic-deriving, we are ready to explore the changes that are necessary to make it support abstraction over multiple parameters.

3.1 Universe

The main change to the universe involves the introduction of a list of parameters, and the separation of fields into a kind of their own. A field can be one of three things.

Constant types (K) are unchanged. Parameters ($P v$) now take an argument $v :: Nat$ to indicate which of the parameters it is. The kind Nat encodes Peano-style natural numbers; we will write 0 for Ze , 1 for $Su Ze$, etc. A generalised form of application ($:@$) replaces R and the old $:@$, encoding the application of a type of kind κ to a type-level list of fields. As we only deal with parameters of kind \star (see Section 4.1), the first argument to $:@$ will always have a kind of the form $\star \rightarrow \dots \rightarrow \star$, and the second argument will have as many elements as necessary to fully saturate the first argument. The interpretation now takes a list of parameters ρ instead of a single parameter. A separate $InField v \rho$ datatype interprets a field representation type v with parameters ρ . Constants are interpreted as before. Parameters are looked-up in the parameter list with a type-level lookup operator $!:$ akin to the value-level $!$ operator. An application $\sigma :@: \chi$ is interpreted by applying σ to each of the arguments in χ , after transforming these (with $ExpandField$) into types of kind \star :

```

kind Univ =
  U
  | F Field
  | Univ :+: Univ
  | Univ :×: Univ

kind Field =
  K  $\star$ 
  | P Nat
  |  $\forall \kappa. \kappa :@: [Field]$ 

data In (v :: Univ) (ρ :: [ $\star$ ]) ::  $\star$  where
  U :: In U ρ
  F :: InField v ρ → In (F v) ρ
  L :: In α ρ → In (α :+: β) ρ
  R :: In β ρ → In (α :+: β) ρ
  :×: :: In α ρ → In β ρ → In (α :×: β) ρ

data InField (v :: Field) (ρ :: [ $\star$ ]) ::  $\star$  where
  K :: α → InField (K α) ρ
  P :: ρ !: v → InField (P v) ρ
  A :: AppFields σ χ ρ → InField (σ :@: χ) ρ

kind Nat = Ze | Su Nat
(ρ :: [ $\star$ ]) !: (v :: Nat) ::  $\star$ 
(α !: ρ) !: Ze = α
(α !: ρ) !: (Su v) = ρ !: v
AppFields σ χ ρ = σ $: ExpandField χ ρ
(σ :: κ) $: (ρ :: [ $\star$ ]) ::  $\star$ 
σ $: '[] = σ
σ $: (α !: β) = (σ α) $: β
ExpandField (χ :: [Field]) (ρ :: [ $\star$ ]) :: [ $\star$ ]
ExpandField '[] ρ = '[]
ExpandField ((K α) !: χ) ρ = α !: ExpandField χ ρ
ExpandField ((P v) !: χ) ρ = (ρ !: v) !: ExpandField χ ρ
ExpandField ((σ :@: ω) !: χ) ρ = (σ $: ExpandField ω ρ) !: ExpandField χ ρ

```

The $ExpandField$ type function converts a list of $Fields$ into user-defined types (of kind \star). Constants are represented by the type in question, parameters are looked up in the parameter list, and applications are expanded into a fully applied type.

The $Generic$ class to mediate between user datatypes and their representation now has a type function $Pars$ which lists the parameters of the datatype:

```

class Generic ( $\alpha :: \star$ ) where
  Rep  $\alpha :: Univ$ 
  Pars  $\alpha :: [\star]$ 
  from ::  $\alpha \rightarrow In (Rep \alpha) (Pars \alpha)$ 
  to    ::  $In (Rep \alpha) (Pars \alpha) \rightarrow \alpha$ 

```

3.2 Datatype encodings

To better understand the universe of our generalised generic-deriving, we show several example datatype encodings in this section. We begin with lists; their encoding is similar to that of Section 2.2, only now we use $(:@:)$ instead of R :

```

instance Generic [ $\alpha$ ] where
  Rep [ $\alpha$ ] =  $U :+ : F (P 0) : \times : F ([] :@: '[P 0])$ 
  Pars [ $\alpha$ ] =  $'[\alpha]$ 
  from [] =  $LU$ 
  from ( $h : t$ ) =  $R (F (P h) : \times : F (A t))$ 

```

The *RTree* type of Section 2.2 can also still be encoded. In fact, since we use the type family *ExpandField* to transform fields v of kind *Field* into \star , instead of using the interpretation *InField* v recursively, we no longer need to use *fmap*; the arguments can be used directly, simplifying the implementation of *from* and *to*:

```

instance Generic (RTree  $\alpha$ ) where
  Rep (RTree  $\alpha$ ) =  $F (P 0) : \times : F ([] :@: '[RTree :@: '[P 0]])$ 
  Pars (RTree  $\alpha$ ) =  $'[\alpha]$ 
  from (RTree  $x xs$ ) =  $F (P x) : \times : F (A xs)$ 

```

Having support for multiple parameters, we can now deal with pairs properly, for example. These are simply a product between two fields with a parameter each:

```

instance Generic ( $\alpha, \beta$ ) where
  Rep ( $\alpha, \beta$ ) =  $F (P 0) : \times : F (P 1)$ 
  Pars ( $\alpha, \beta$ ) =  $'[\alpha, \beta]$ 
  from ( $a, b$ ) =  $F (P a) : \times : F (P b)$ 

```

A more complicated example shows how we can mix datatypes with a different number of parameters, and partially instantiated datatypes:

```

data D  $\alpha \beta \gamma = D \beta [(\alpha, Int)] (RTree [\gamma])$ 
instance Generic (D  $\alpha \beta \gamma$ ) where
  Rep (D  $\alpha \beta \gamma$ ) =
     $F (P 1) : \times : F ([] :@: '[(,) :@: '[P 0, K Int]]) : \times : F (RTree :@: '[[] :@: '[P 2]])$ 
  Pars (D  $\alpha \beta \gamma$ ) =  $'[\alpha, \beta, \gamma]$ 
  from (D  $a b c$ ) =  $F (P a) : \times : F (A b) : \times : F (A c)$ 

```

Even nested datatypes can be encoded, as our application supports the notion of recursion with parameters instantiated to other applications:

```

data Perfect  $\alpha = \text{Perfect} (\text{Perfect} (\alpha, \alpha)) \mid \text{End } \alpha$ 
instance Generic (Perfect  $\alpha$ ) where
  Rep (Perfect  $\alpha$ ) = F (Perfect :@: ‘[(,) :@: ‘[P 0, P 0]]’ :+ : F (P 0))
  Pars (Perfect  $\alpha$ ) = ‘[ $\alpha$ ]’
  from (Perfect  $x$ ) = L (F (A  $x$ ))
  from (End  $x$ )    = R (F (P  $x$ ))

```

As we have seen, our generalisation of generic-deriving supports all the previous datatypes, plus many new ones, involving any number of parameters of kind \star .

3.3 Mapping

While designing a new or improved generic programming library, defining the universe and showing example datatype encodings is not enough; it is, in fact, easy to define a simple and encompassing universe that is then found not to be suitable for defining generic functions. As such, we show that our universe allows defining a generalised map function, which maps n different functions over n datatype parameters.

Preliminaries As the generalised map takes n functions, one per datatype parameter, we need a value-level counterpart to our type-level lists of parameters. We use a strongly typed heterogeneous list [4] for this purpose:

```

data HList ( $\rho :: [\star]$ ) where
  HNil   :: HList ‘[]’
  HCons ::  $\alpha \rightarrow \text{HList } \beta \rightarrow \text{HList } (\alpha ‘:’ \beta)$ 

```

We also need a way to pick the n -th element from such a list; we define a *Lookup* type class for this purpose, as the function is defined by induction on the type-level natural numbers. We use a *Proxy* to carry type information at the value-level (the index being looked up):

```

data Proxy ( $\sigma :: \kappa$ ) = Proxy
class Lookup ( $\rho :: [\star]$ ) ( $v :: \text{Nat}$ ) where
  lookup :: Proxy  $v \rightarrow \text{HList } \rho \rightarrow \rho !: v$ 
instance Lookup  $\rho$  Ze where
  lookup _ (HCons  $f$  _) =  $f$ 
instance (Lookup  $\beta$   $v$ )  $\Rightarrow$  Lookup ( $\alpha ‘:’ \beta$ ) (Su  $v$ ) where
  lookup _ (HCons _  $fs$ ) = lookup (Proxy :: Proxy  $v$ )  $fs$ 

```

User-facing type class We are now ready to see the class that generalises *Functor* of Section 2.3. This is a multi-parameter type class, taking a parameter σ for the unsaturated type we are mapping over, and a list of function types τ encoding the types of the functions we will be mapping. As σ will often be ambiguous in the code for the generic definition of map, we provide a version *gmap_P* with an extra argument that serves to identify the σ . A version without the proxy (*gmap*) is provided with a default for convenient usage:


```

class GMap ( $\sigma :: \kappa$ ) ( $\tau :: [*]$ ) |  $\tau \rightarrow \kappa$  where
  gmap :: HList  $\tau \rightarrow \sigma$  :$ Doms  $\tau \rightarrow \sigma$  :$ Codoms  $\tau$ 
  gmap = gmapP (Proxy :: Proxy  $\sigma$ )
  gmapP :: Proxy  $\sigma \rightarrow$  HList  $\tau \rightarrow \sigma$  :$ Doms  $\tau \rightarrow \sigma$  :$ Codoms  $\tau$ 

```

Our generalised map, *gmap*, takes two arguments. The first is an *HList* of functions to map. The second is the type σ applied to the domains of the functions we are mapping. Its return type is again σ , but now applied to the codomains of the same functions. For example, for lists, σ is `[]`, and τ is `'[$\alpha \rightarrow \beta$]`. In this case, *gmap* gets the expected type *HList* `'[$\alpha \rightarrow \beta$] \rightarrow [α] \rightarrow [β]`. The functional dependency $\tau \rightarrow \kappa$ is essential to prevent ambiguity when using *gmap*—and indeed the kind of σ is uniquely determined by the (length of the) list τ . The type functions *Doms* and *Codoms* compute the domains and codomains of the list of functions τ :

```

Doms ( $\tau :: [*]$ ) :: [*]
Doms '[] = '[]
Doms (( $\alpha \rightarrow \beta$ ) ' :  $\tau$ ) =  $\alpha$  ' : Doms  $\tau$ 
Codoms ( $\tau :: [*]$ ) :: [*]
Codoms '[] = '[]
Codoms (( $\alpha \rightarrow \beta$ ) ' :  $\tau$ ) =  $\beta$  ' : Codoms  $\tau$ 

```

Mapping on the generic representation We have seen *GMap*, the user-facing type class for the generalised map. Its counterpart for representation types is *GMap_R*, which operates on the interpretation of a representation:

```

class GMapR ( $v :: Univ$ ) ( $\tau :: [*]$ ) where
  gmapR :: HList  $\tau \rightarrow In$   $v$  (Doms  $\tau$ )  $\rightarrow In$   $v$  (Codoms  $\tau$ )

```

We now go through the instances of *GMap_R* for each representation type. There is nothing to map over in units, and sums and products are simply traversed through:

```

instance GMapR U  $\tau$  where
  gmapR _ U = U
instance (GMapR  $\alpha$   $\tau$ , GMapR  $\beta$   $\tau$ )  $\Rightarrow$  GMapR ( $\alpha$  :+  $\beta$ )  $\tau$  where
  gmapR fs (L x) = L (gmapR fs x)
  gmapR fs (R x) = R (gmapR fs x)
instance (GMapR  $\alpha$   $\tau$ , GMapR  $\beta$   $\tau$ )  $\Rightarrow$  GMapR ( $\alpha$  : $\times$   $\beta$ )  $\tau$  where
  gmapR fs (x : $\times$  y) = gmapR fs x : $\times$  gmapR fs y

```

Fields require more attention, so we define a separate type class *GMap_{RF}* to handle them:

```

instance (GMapRF  $v$   $\tau$ )  $\Rightarrow$  GMapR (F  $v$ )  $\tau$  where
  gmapR fs (F x) = F (gmapRF fs x)
class GMapRF ( $v :: Field$ ) ( $\tau :: [*]$ ) where
  gmapRF :: HList  $\tau \rightarrow InField$   $v$  (Doms  $\tau$ )  $\rightarrow InField$   $v$  (Codoms  $\tau$ )

```

Constants, however, just like units, are returned unchanged:

instance $GMap_{RF} (K \alpha) \tau$ **where**
 $gmap_{RF} _ (K x) = K x$

For a parameter $P v$, we need to lookup the right function to map over. We again use a separate type class, $GMap_{RP}$, and we traverse the input list of functions until we reach the v -th function. We thus require the list of functions τ to have its elements in the same order as the datatype parameters:

instance $(GMap_{RP} v \tau) \Rightarrow GMap_{RF} (P v) \tau$ **where**
 $gmap_{RF} fs (P x) = P (gmap_{RP} (Proxy :: Proxy v) fs x)$

class $GMap_{RP} (v :: Nat) (\tau :: [*])$ **where**
 $gmap_{RP} :: Proxy v \rightarrow HList \tau \rightarrow (Doms \tau) !: v \rightarrow (Codoms \tau) !: v$

instance $GMap_{RP} Ze ((\alpha \rightarrow \beta) ' : \tau)$ **where**
 $gmap_{RP} _ (HCons f _) x = f x$

instance $(GMap_{RP} v \tau) \Rightarrow GMap_{RP} (Su v) ((\alpha \rightarrow \beta) ' : \tau)$ **where**
 $gmap_{RP} _ (HCons _ fs) p = gmap_{RP} (Proxy :: Proxy v) fs p$

Handling application The only representation type we still have to deal with is $(:@)$. This is also the most challenging case. As a running example, consider the second field of the D datatype of Section 3.2. D has three parameters, α , β , and γ , and the second field of its only constructor has type $[(\alpha, Int)]$, represented as $[] :@: '[(\alpha, Int) :@: '[P 0, K Int]]$. In this situation, we intend to transform $[(\alpha, Int)]$ into $[(\alpha', Int)]$, having available a function of type $\alpha \rightarrow \alpha'$. We do this by requiring the availability of $gmap$ for this particular list type; that is, we require an instance $GMap [] '[(\alpha, Int) \rightarrow (\alpha', Int)]$. Having such an instance, we can simply $gmap$ over the argument. However, the functions we map need to be adapted to this new argument. That is the task of $AdaptFs$, which we explain below.

A type function $MakeFs$ computes the types of the functions to be mapped inside the applied type. For example, $MakeFs '[(\alpha, Int) :@: '[P 0, K Int]] '[\alpha \rightarrow \alpha', \beta \rightarrow \beta', \gamma \rightarrow \gamma']$ is $'[(\alpha, Int) \rightarrow (\alpha', Int)]$. We use proxies to fix otherwise ambiguous types when invoking $gmap_P$ and $adaptFs$:

$MakeFs (\rho :: [Field]) (\tau :: [*]) :: [*]$
 $MakeFs [] \tau = []$
 $MakeFs ((K \alpha) ' : \rho) \tau = (\alpha \rightarrow \alpha) ' : MakeFs \rho \tau$
 $MakeFs ((P v) ' : \rho) \tau = (\tau !: v) ' : MakeFs \rho \tau$
 $MakeFs ((\sigma :@: \omega) ' : \rho) \tau =$
 $(AppFields \sigma \omega (Doms \tau) \rightarrow AppFields \sigma \omega (Codoms \tau)) ' : MakeFs \rho \tau$

instance $(GMap \sigma (MakeFs \rho \tau)$
 $, AdaptFs \rho \tau$
 $, ExpandField \rho (Doms \tau) \sim Doms (MakeFs \rho \tau)$
 $, ExpandField \rho (Codoms \tau) \sim Codoms (MakeFs \rho \tau)$
 $) \Rightarrow GMap_{RF} (\sigma :@: \rho) \tau$ **where**
 $gmap_{RF} fs (A x) = A (gmap_P p_1 (adaptFs p_2 fs) x)$

where $p_1 = \text{Proxy}::\text{Proxy } \sigma$
 $p_2 = \text{Proxy}::\text{Proxy } \rho$

This instance has four constraints, two of them being equality constraints, introduced by the \sim operator: a constraint of the form $\alpha \sim \beta$ indicates that the type α must be equal to the type β . The four constraints of this instance are: the ability to map over the argument type, the requirement to rearrange the functions we map over, and two coherence conditions on the behaviour of *ExpandField* and *MakeFs* over the list of functions. The latter are always true for valid ρ and τ .

We are left with the task of adapting the functions to be mapped over. In our running example, we have an *HList* $[\alpha \rightarrow \alpha', \beta \rightarrow \beta', \gamma \rightarrow \gamma']$, and we have to produce an *HList* $[(\alpha, \text{Int}) \rightarrow (\alpha', \text{Int})]$. This is done by *adaptFs*, a method of the type class *AdaptFs* which is implemented by induction on the list of fields to be mapped over:

class *AdaptFs* ($\rho :: [\text{Field}]$) ($\tau :: [*]$) **where**
 $\text{adaptFs} :: \text{Proxy } \rho \rightarrow \text{HList } \tau \rightarrow \text{HList } (\text{MakeFs } \rho \tau)$

For an empty list of arguments, we return an empty list of functions. If the argument is a constant, we ignore it and proceed to the next argument. For a parameter $P v$, we use the v -th function of the original list of functions, and proceed to the next argument:

instance *AdaptFs* $[\] \tau$ **where**
 $\text{adaptFs } _ = \text{HNil}$
instance (*AdaptFs* $\rho \tau$) \Rightarrow *AdaptFs* ($(K \alpha) ':\rho$) τ **where**
 $\text{adaptFs } _ fs = \text{HCons } id (\text{adaptFs } (\text{Proxy}::\text{Proxy } \rho) fs)$
instance (*AdaptFs* $\rho \tau, \text{Lookup } \tau v$) \Rightarrow *AdaptFs* ($(P v) ':\rho$) τ **where**
 $\text{adaptFs } _ fs = \text{HCons } (\text{lookup } p_1 fs) (\text{adaptFs } p_2 fs)$
where $p_1 = \text{Proxy}::\text{Proxy } v$
 $p_2 = \text{Proxy}::\text{Proxy } \rho$

The most delicate case is, again, application. Back to our running example, this is where we have to produce a function of type $(\alpha, \text{Int}) \rightarrow (\alpha', \text{Int})$. We do so by requiring an instance *GMap* $(,)$ $[\alpha \rightarrow \alpha', \text{Int} \rightarrow \text{Int}]$, reusing *MakeFs* and *AdaptFs* in the process. We also need to proceed recursively for the rest of the arguments. Again, we have the same two coherence conditions on the behaviour of *ExpandField* and *MakeFs* over the list of functions, and use proxies to fix ambiguous types:

instance (*GMap* σ (*MakeFs* $\omega \tau$),
AdaptFs $\omega \tau$,
AdaptFs $\rho \tau$,
 $\text{ExpandField } \omega (\text{Doms } \tau) \sim \text{Doms } (\text{MakeFs } \omega \tau)$,
 $\text{ExpandField } \omega (\text{Codoms } \tau) \sim \text{Codoms } (\text{MakeFs } \omega \tau)$)
 \Rightarrow *AdaptFs* ($(\sigma :@: \omega) ':\rho$) τ **where**
 $\text{adaptFs } _ fs = \text{HCons } (\text{gmap}_P p_1 (\text{adaptFs } p_2 fs)) (\text{adaptFs } p_3 fs)$
where $p_1 = \text{Proxy}::\text{Proxy } \sigma$
 $p_2 = \text{Proxy}::\text{Proxy } \omega$
 $p_3 = \text{Proxy}::\text{Proxy } \rho$

The generic definition of the generalised map is thus complete, and ready to be used.

3.4 Example usage

Before we instantiate `map` to our example datatypes of Section 3.2, we first provide a generic default [11] to make instantiation simpler. This default, for the `GMap` class, will allow us to then give empty instances of the class, which will automatically use the generic definition for the generalised map. The default version of `gmapP` converts a value into its generic representation, applies `gmapR`, and then converts back to a user datatype. This requires several constraints (which would all be inferred if the function was defined at the top level). First, we introduce α and β as synonyms for the input and output type, respectively, for mere convenience. In the case of the `GMap [] '[$\gamma \rightarrow \gamma'$]` instance, for example, α is $[\gamma]$, and β is $[\gamma']$. We also require a `Generic [γ]` instance (for *from*), and a `Generic [γ']` instance (for *to*); these instances will coincide, and indeed we also require that the representation `Rep [γ]` be the same as `Rep [γ']` (which is the case). Furthermore, the parameters of $[\gamma]$ have to coincide with the domains of $'[\gamma \rightarrow \gamma']$, and the parameters of $[\gamma']$ have to coincide with the codomains of the same list. Finally, we require the ability to map generically over the representation of the input list:

```

default gmapP :: ( $\alpha \sim (\sigma \text{ :\$ } \text{Doms } \tau)$ 
  ,  $\beta \sim (\sigma \text{ :\$ } \text{Codoms } \tau)$ 
  , Generic  $\alpha$ , Generic  $\beta$ 
  , Rep  $\alpha \sim \text{Rep } \beta$ 
  , Pars  $\alpha \sim \text{Doms } \tau$ 
  , Pars  $\beta \sim \text{Codoms } \tau$ 
  , GMapR (Rep  $\alpha$ )  $\tau$ 
  )  $\Rightarrow \text{Proxy } \sigma \rightarrow \text{HList } \tau \rightarrow \alpha \rightarrow \beta$ 
gmapP _fs = to  $\circ$  gmapR fs  $\circ$  from

```

With this default in place, we are ready to instantiate our example datatypes:

```

instance GMap [] '[ $\alpha \rightarrow \alpha'$ ]
instance GMap RTree '[ $\alpha \rightarrow \alpha'$ ]
instance GMap (,) '[ $\alpha \rightarrow \alpha'$ ,  $\beta \rightarrow \beta'$ ]
instance GMap D '[ $\alpha \rightarrow \alpha'$ ,  $\beta \rightarrow \beta'$ ,  $\gamma \rightarrow \gamma'$ ]
instance GMap Perfect '[ $\alpha \rightarrow \alpha'$ ]

```

Using the generic default, instantiation is kept simple and concise. We can verify that our map works as expected in an example that should cover all the representation types:

```

x :: D Int Float Char
x = D 0.2 [(0,0), (1,1)] (RTree "p" [])
y :: D Int String Char
y = gmap (HCons (+1) (HCons show (HCons (const 'q') HNil))) x

```

Indeed, `y` evaluates to `D "0.2" [(1,0), (2,1)] (RTree "q" [])` as expected.

4 Limitations and future work

In this section we discuss the limitations of our new generic-deriving, and propose future research directions.

4.1 Parameters of higher kinds

While our generalisation supports any number of parameters of kind \star , it is unable to deal with parameters of higher kinds. Consider the following two datatypes representing generalised trees:

```
data GTree1  $\phi$   $\alpha$  = GTree1  $\alpha$  ( $\phi$  (GTree1  $\phi$   $\alpha$ ))
data GTree2  $\alpha$   $\phi$  = GTree2  $\alpha$  ( $\phi$  (GTree2  $\alpha$   $\phi$ ))
```

The most general mapping function for *GTree₁* has the following type:

$$(\alpha \rightarrow \beta) \rightarrow (\forall \alpha \beta. (\alpha \rightarrow \beta) \rightarrow \phi \alpha \rightarrow \psi \beta) \rightarrow \text{GTree}_1 \phi \alpha \rightarrow \text{GTree}_1 \psi \beta$$

The generalisation of this paper provides a map of type $(\alpha \rightarrow \beta) \rightarrow \text{GTree}_1 \phi \alpha \rightarrow \text{GTree}_1 \phi \beta$, therefore ignoring the ϕ parameter of kind $\star \rightarrow \star$. For *GTree₂*, however, our approach cannot even provide that simple map; we cannot give the *GMap* instance, since the parameters of kind \star of interest come before a parameter of kind $\star \rightarrow \star$, and the second parameter of *GMap* is a list of kind $[\star]$ (so all arguments must have kind \star).

As such, we support generic abstraction only over the parameters of kind \star which come after any parameters of other kinds. Lifting this restriction is not trivial. Recall that the *Pars* type family has return kind $[\star]$. To generalise to multiple parameters, we should make this return kind be a (promoted) heterogeneous list. This is currently not possible, as heterogeneous lists are implemented as GADTs, which cannot be promoted. Foregoing giving *Pars* the correct kind and working with nested tuples instead gives rise to many kind ambiguities, which are cumbersome to solve. As such, we hope that the promotion mechanism is extended to allow promotion of GADTs soon [15], and defer generalising our approach to parameters of arbitrary kinds until then.

4.2 Integration with existing generic programming libraries

The introduction of our new generic-deriving raises the question of how to upgrade from the old version. Our changes are not backwards compatible, but since the new version encodes strictly more information than the previous one, we can provide a conversion that automatically transforms the new representation into the old one, therefore remaining compatible with old code [10]. The core of this conversion is a type-level function $D_{n \rightarrow D}$ that converts the new representation into the old one. We show a prototype implementation here, subscripting the new generic-deriving codes with an n to distinguish them from the old ones. Units, sums, and products are converted trivially. Fields are handled by a separate function $D_{n \rightarrow D_F}$. Constants are converted trivially. For a parameter, we return P if it is the last parameter of the datatype (the only one supported by the old version of generic-deriving), or a constant otherwise. We make use of type-level if-then-else, equality on *Nat*, and length. Applications of types of kind $\star \rightarrow \star$ are converted into compositions. For applications of types of higher arity, we first apply the type to all its arguments but the last:

```
IfThenElse ( $\alpha :: \text{Bool}$ ) ( $\beta :: \kappa$ ) ( $\gamma :: \kappa$ ) ::  $\kappa$ 
IfThenElse True  $\beta$   $\gamma = \beta$ 
IfThenElse False  $\beta$   $\gamma = \gamma$ 
```

$$\begin{aligned}
(\alpha :: \text{Nat}) &\equiv_{\text{Nat}} (\beta :: \text{Nat}) :: \text{Bool} \\
Ze &\equiv_{\text{Nat}} Ze = \text{True} \\
Su \ v &\equiv_{\text{Nat}} Ze = \text{False} \\
Ze &\equiv_{\text{Nat}} Su \ v = \text{False} \\
Su \ v &\equiv_{\text{Nat}} Su \ v = v \equiv_{\text{Nat}} v \\
\text{Length } (\rho :: [\kappa]) &:: \text{Nat} \\
\text{Length } [] &= Ze \\
\text{Length } (\alpha \text{ ' : } \rho) &= Su (\text{Length } \rho) \\
D_{n \rightarrow D} (v :: \text{Univ}_n) (\rho :: [\star]) &:: \text{Univ} \\
D_{n \rightarrow D} U_n \quad \rho &= U \\
D_{n \rightarrow D} (\alpha \text{ :+}_n \beta) \rho &= D_{n \rightarrow D} \alpha \rho \text{ :+} : D_{n \rightarrow D} \beta \rho \\
D_{n \rightarrow D} (\alpha \text{ :}\times_n \beta) \rho &= D_{n \rightarrow D} \alpha \rho \text{ :}\times : D_{n \rightarrow D} \beta \rho \\
D_{n \rightarrow D} (F_n \alpha) \quad \rho &= D_{n \rightarrow D_F} \alpha \rho \\
D_{n \rightarrow D_F} (v :: \text{Field}_n) (\rho :: [\star]) &:: \text{Univ} \\
D_{n \rightarrow D_F} (K_n \alpha) &\quad \rho = K \alpha \\
D_{n \rightarrow D_F} (P_n v) &\quad \rho = \text{IfThenElse } (v \equiv_{\text{Nat}} \text{Length } \rho) P (K (\rho \text{ !: } v)) \\
D_{n \rightarrow D_F} (\phi \text{ :@}_n \text{ ' } [\alpha]) &\quad \rho = \phi \text{ :@} : D_{n \rightarrow D_F} \alpha \rho \\
D_{n \rightarrow D_F} (\phi \text{ :@}_n \text{ : } (\alpha \text{ ' : } \beta \text{ ' : } \gamma)) &\rho = D_{n \rightarrow D_F} ((\text{AppFields } \phi \text{ ' } [\alpha] \rho) \text{ :@}_n \text{ : } (\beta \text{ ' : } \gamma)) \rho
\end{aligned}$$

The introduction of yet another generic programming library underscores the need for a single mechanism for type reflection baked into the compiler, from which other mechanisms, such as our new generic-deriving, or *Data* and *Typeable* [5], could then be derived.

4.3 Parameter genericity vs. arity genericity

The approach described in this paper allows us to define generic functions that operate over multiple datatype parameters. This is not the same as generic functions that operate at diverse arities [14]. Consider the following generic functions:

$$\begin{aligned}
\text{map}_n^1 &:: \text{HList } [\alpha_1^1, \dots, \alpha_n^1] \rightarrow \phi (\alpha_1^1 \dots \alpha_n^1) \\
\text{map}_n^2 &:: \text{HList } [\alpha_1^1 \rightarrow \alpha_1^2, \dots, \alpha_n^1 \rightarrow \alpha_n^2] \rightarrow \phi (\alpha_1^1 \dots \alpha_n^1) \rightarrow \phi (\alpha_1^2 \dots \alpha_n^2) \\
\text{map}_n^3 &:: \text{HList } [\alpha_1^1 \rightarrow \alpha_1^2 \rightarrow \alpha_1^3, \dots, \alpha_n^1 \rightarrow \alpha_n^2 \rightarrow \alpha_n^3] \\
&\rightarrow \phi (\alpha_1^1 \dots \alpha_n^1) \rightarrow \phi (\alpha_1^2 \dots \alpha_n^2) \rightarrow \phi (\alpha_1^3 \dots \alpha_n^3) \\
\text{map}_n^m &:: \text{HList } [\alpha_1^1 \rightarrow \dots \rightarrow \alpha_1^m, \dots, \alpha_n^1 \rightarrow \dots \rightarrow \alpha_n^m] \\
&\rightarrow \phi (\alpha_1^1 \dots \alpha_n^1) \rightarrow \dots \rightarrow \phi (\alpha_1^m \dots \alpha_n^m)
\end{aligned}$$

The function map_n^1 , or *repeat*, creates a ϕ -structure given elements for the parameter positions. The function map_n^2 , equivalent to the generic *gmap* of Section 3.3, is the function that maps over a ϕ -structure, taking one function per parameter. The function map_n^3 , or *zipWith*, is the function that takes two ϕ -structures and zips them when their shapes are compatible, using the provided functions to zip the parameters. The function map_n^m is the generalisation of the previous three, in the arity-generic sense. Our approach allows defining each of map_n^1 , map_n^2 , map_n^3 , etc., individually, as separate generic functions. It does not allow defining map_n^m ; that generalisation is described by Weirich and Casingh-

ino [14], in the dependently-typed programming language Agda. It remains to see how to transfer the concept of arity-genericity to Haskell.

5 Conclusion

In this paper we have seen how we can use the promotion mechanism together with kind polymorphism to encode a generic representation of datatypes that supports abstraction over multiple parameters (of kind \star). This enables a whole new class of generic functionality: we have shown a generalised map, but also folding, traversing, and zipping, for example, are now possible. We plan to implement support for the new generic-deriving in GHC soon, so that users can take advantage of the new functionality without needing to write their own *Generic* instances.

Acknowledgements This work was supported by the EP/J010995/1 EPSRC grant. Jeremy Gibbons and anonymous referees provided valuable feedback on an early draft of this paper.

Bibliography

- [1] Artem Alimarine and Rinus Plasmeijer. A generic programming extension for Clean. In *The 13th International Workshop on the Implementation of Functional Languages*, volume 2312 of *Lecture Notes in Computer Science*, pages 168–185. Springer, 2001. doi:10.1007/3-540-46028-4_11.
- [2] Roland Backhouse, Patrik Jansson, Johan Jeuring, and Lambert Meertens. Generic programming: An introduction. In Doaitse S. Swierstra, Pedro R. Henriques, and José Nuno Oliveira, editors, *3rd International Summer School on Advanced Functional Programming*, volume 1608, pages 28–115. Springer-Verlag, 1999. doi:10.1007/10704973_2.
- [3] Ralf Hinze, Johan Jeuring, and Andres Löf. Comparing approaches to generic programming in Haskell. In Roland Backhouse, Jeremy Gibbons, Ralf Hinze, and Johan Jeuring, editors, *Datatype-Generic Programming*, volume 4719 of *Lecture Notes in Computer Science*, pages 72–149. Springer-Verlag, 2007. doi:10.1007/978-3-540-76786-2_2.
- [4] Oleg Kiselyov, Ralf Lämmel, and Kean Schupke. Strongly typed heterogeneous collections. In *Proceedings of the 2004 ACM SIGPLAN Workshop on Haskell, Haskell '04*, pages 96–107. ACM, 2004. doi:10.1145/1017472.1017488.
- [5] Ralf Lämmel and Simon Peyton Jones. Scrap your boilerplate: a practical design pattern for generic programming. In *Proceedings of the 2003 ACM SIGPLAN International Workshop on Types in Languages Design and Implementation*, pages 26–37. ACM, 2003. doi:10.1145/604174.604179.
- [6] Andres Löf. *Exploring Generic Haskell*. PhD thesis, Universiteit Utrecht, 2004. <http://igitur-archive.library.uu.nl/dissertations/2004-1130-111344>.
- [7] Andres Löf and José Pedro Magalhães. Generic programming with indexed functors. In *Proceedings of the 7th ACM SIGPLAN Workshop on Generic Programming*, pages 1–12. ACM, 2011. doi:10.1145/2036918.2036920.

- [8] José Pedro Magalhães. *Less Is More: Generic Programming Theory and Practice*. PhD thesis, Universiteit Utrecht, 2012.
- [9] José Pedro Magalhães. The right kind of generic programming. In *Proceedings of the 8th ACM SIGPLAN Workshop on Generic Programming, WGP '12*, pages 13–24, New York, NY, USA, 2012. ACM. ISBN 978-1-4503-1576-0. doi:10.1145/2364394.2364397.
- [10] José Pedro Magalhães and Andres Löf. Generic generic programming, 2014. URL <http://dreixel.net/research/pdf/ggp.pdf>. Accepted for publication at Practical Aspects of Declarative Languages (PADL'14).
- [11] José Pedro Magalhães, Atze Dijkstra, Johan Jeuring, and Andres Löf. A generic deriving mechanism for Haskell. In *Proceedings of the 3rd ACM Haskell Symposium on Haskell*, pages 37–48. ACM, 2010. doi:10.1145/1863523.1863529.
- [12] Simon Peyton Jones, editor. *Haskell 98, Language and Libraries. The Revised Report*. Cambridge University Press, 2003. doi:10.1017/S0956796803000315. Journal of Functional Programming Special Issue 13(1).
- [13] Alexey Rodriguez Yakushev, Johan Jeuring, Patrik Jansson, Alex Gerdes, Oleg Kiselyov, and Bruno C.d.S. Oliveira. Comparing libraries for generic programming in Haskell. In *Proceedings of the 1st ACM SIGPLAN Symposium on Haskell*, pages 111–122. ACM, 2008. doi:10.1145/1411286.1411301.
- [14] Stephanie Weirich and Chris Casinghino. Arity-generic datatype-generic programming. In *Proceedings of the 4th ACM SIGPLAN Workshop on Programming Languages meets Program Verification*, pages 15–26. ACM, 2010. doi:10.1145/1707790.1707799.
- [15] Stephanie Weirich, Justin Hsu, and Richard A. Eisenberg. System FC with explicit kind equality. In *Proceedings of the 18th ACM SIGPLAN International Conference on Functional Programming, ICFP '13*, pages 275–286. ACM, 2013. doi:10.1145/2500365.2500599.
- [16] Brent A. Yorgey, Stephanie Weirich, Julien Cretin, Simon Peyton Jones, Dimitrios Vytiniotis, and José Pedro Magalhães. Giving Haskell a promotion. In *Proceedings of the 8th ACM SIGPLAN Workshop on Types in Language Design and Implementation*, pages 53–66. ACM, 2012. doi:10.1145/2103786.2103795.